A Survey of Phrase Projectivity in *Antigone*

Jonathan Sterling

April 2013

THIS IS A DRAFT. It is currently lacking a bibliography, and the analysis is a bit shallow currently.

In this paper, I will show how (and to what degree) phrase projectivity corresponds with register and meter in Sophocles's *Antigone*, by developing a quantitative metric for projectivity and comparing it across lyrics, trimeters and anapaests.

1 Dependency Trees and Their Projectivity

A dependency tree encodes the head-dependent relation for a string of words, where arcs are drawn from heads to their dependents. We consider a phrase *projective* when these arcs do not cross each other, and *discontinuous* to the extent that any of the arcs intersect. Figure 1 is a minimal pair that demonstrates how hyperbaton introduces a projectivity violation; in this case, a path of dependency "wraps around itself".



(a) "Full of plentiful supplies" (Xenophon, *Anabasis* 3.5.1) is fully projective.

(b) "Full of many good things" (Plato, *Laws* 906a) has one projectivity violation.

Figure 1: Examples drawn from Devine & Stephens.

In addition to the above, adjacent phrases (that is, phrases at the same level in the tree) may interlace, causing projectivity violations. This is commonly introduced by Wackernagel's Law, as in Figure 2, where the placement of μὲν δὴ interlaces with the τὰ...πόλεος phrase.

We consider a violation to have occured for each and every intersection of lines on such a drawing; thus, the hyperbaton of one word may introduce multiple violations.



Figure 2: "[The gods] righted the matters of the city..." (*Antigone* 162–163) has one projectivity violation, due to the μὲν δὴ falling in Wackernagel's Position.

Consider, for instance, Figure 3, in which five violations are brought about by the displacement of φονώσαισιν. In this way, the number of intersections is a good heuristic for judging the severity of hyperbata.



Figure 3: "And he stood over the rooftops, gaped in a circle with murderous spears around the seven-gated mouth, and left" (*Antigone* 117–120) has six projectivity violations.

1.1 Counting Violations

Drawing trees and counting intersections is time-consuming and error-prone, especially since the number of intersections may vary if one is not consistent with the relative height of arcs. It is clear, then, that a computer ought to be able to do the job faster and more accurately than a human, given at least the head-dependent relations for a corpus.

The formal algorithm for counting the number of intersections is given in Appendix A, but I shall reproduce an informal and mostly nontechnical version of it here. First, we index each word in the sentence by its linear position, and cross-reference it with the linear position of its head:

στὰς δ' *ὑπὲρ* *μελάθρων* *φονώσαισιν* *ἀμφικανὼν* *κύκλῳ* *λόγχαις*
1:11 2:11 3:1 4:3 5:8 6:11 7:6 8:6
ἑπτάπυλον *στόμα* *ἔβα*
9:10 10:6 11:_

Next, arrange the dependencies into a tree as in Figure 4. Then, counting upwards from the lowest edges (i.e. the lines) in the tree up to the topmost ones, make a list of edges indexed by vertical level as in Table 1.



Figure 4: The dependency relations arranged in a non-linear tree.

level	*edges*
1	$3 \leftrightarrow 4, 5 \leftrightarrow 8, 9 \leftrightarrow 10$
2	$1 \leftrightarrow 3, 6 \leftrightarrow 7, 6 \leftrightarrow 8, 6 \leftrightarrow 10$
3	$1 \leftrightarrow 11, 2 \leftrightarrow 11, 6 \leftrightarrow 11$

Table 1: Edges of the tree in Figure 4 arranged by level.

Then, each edge in our table must be checked for violations against all the other edges in the table except those which are in a level higher than it. The level of the edge corresponds with the height at which we drew the arcs; this condition arises out of the fact that an arc cannot cross an arc that is above it, rather, only one that is below it.

Next, we must figure out all the possible ways for an arc to intersect another at given levels. These are enumerated in detail in the function checkEdges in Appendix A.2, but suffice it to say that they fall into a few main categories:

1. Both vertices of the higher edge are within the bounds of the lower edge. This is a double violation, as both sides of an arc will extrude through another.

2. One vertex of the higher edge is within the bounds of the lower edge, and the other vertex is not; this vertex is allowed to be equal to the second vertex of the lower edge. In either case, this is a single violation, as just one intersection occurs.

3. The edges are at the same level, and one vertex of the higher edge is neither within bounds of the other, nor equal to any of the vertices of the other.

Using this procedure, we shall have found the edge violations which are listed in Table 2, which are 6 in total.

$2 \leftrightarrow 11$	$1 \leftrightarrow 3$	1
$6 \leftrightarrow 11$	$5 \leftrightarrow 8$	1
$6 \leftrightarrow 7$	$5 \leftrightarrow 8$	2
$6 \leftrightarrow 8$	$5 \leftrightarrow 8$	1
$6 \leftrightarrow 10$	$5 \leftrightarrow 8$	1
		$\mathsf{total} = 6$

Table 2: Projectivity violations which arise from the edges in Table 1.

1.2 ω: a metric of projectivity

In order for our view of a text's overall projectivity to not be skewed by its length, we must have a ratio. For the purposes of this paper, we shall call this metric ω, as given by the following ratio:

$$\omega = \frac{\text{number of violations}}{\text{number of arcs}}$$

2 The Perseus Treebank

The Perseus Ancient Greek Dependency Treebank is a massive trove of annotated texts that encode the all dependency relations in every sentence. The data is given in an XML (E**x**tensible **M**arkup **L**anguage) format resembling the following:

```
<sentence id="2900759">
  <word id="1" form="χρὴ" lemma="χρή" head="0" />
  <word id="2" form="δὲ" lemma="δέ" head="1" />
  ...
</sentence>
```

Every sentence is given a unique, sequential identifier; within each sentence, every word is indexed by its linear position and coreferenced with the linear position of its dominating head. In the case of the data for *Antigone*, the maximal head of each sentence has its own head given as `0`. Appendix B deals with parsing these XML representations into dependency trees for which we can compute ω.

(a) Choral Odes

Lines		ω
$100 \cdots 154$	*First choral ode*	0.77
$332 \cdots 375$	*Second choral ode*	0.47
$583 \cdots 625$	*Third choral ode*	0.53
$781 \cdots 800$	*Fourth choral ode*	0.47
$944 \cdots 987$	*Fifth choral ode*	0.41
$1116 \cdots 1152$	*Sixth choral ode*	0.75
	mean $\omega = 0.57$, sdev $= 0.15$	

(b) Laments

Lines		ω
$806 \cdots 816$	*Antigone's Lament*	1.29
$823 \cdots 833$	*Antigone's Lament (cntd.)*	0.79
$839 \cdots 882$	*Antigone's Lament (cntd.)*	0.54
$1261 \cdots 1269$	*Kreon's Lament*	0.52
$1283 \cdots 1292$	*Kreon's Lament (cntd.)*	0.88
$1306 \cdots 1311$	*Kreon's Lament (cntd.)*	0.33
$1317 \cdots 1325$	*Kreon's Lament (cntd.)*	1.08
$1239 \cdots 1246$	*Kreon's Lament (cntd.)*	0.47
	mean $\omega = 0.74$, sdev $= 0.33$	

Table 3: Lyrics

3 Projectivity in Antigone

To observe the variation of projectivity within a text, then, one may make a selection of sentences that have something in common, compute their trees and thence derive ω, and then average the results. Then that quantity may be compared with that of other selections.

I have chosen to compare projectivity in lyrics, anapaests and trimeters. Lyrics I have divided into two categories: choral odes and laments, whereas I divide trimeters into medium-to-long speeches and stichomythia.

To that end, I have selected passages from *Antigone* and organized them by type. Table 3 enumerates the lyric passages of the play, along with their computed mean ω values, and a final mean of means with the standard deviation of the set. Table 4 does

Lines		ω
$155\cdots161$	*Kreon's Entrance*	0.33
$376\cdots383$	*Antigone's Entrance*	0.64
$526\cdots530$	*Ismene's Entrance*	0.05
$626\cdots630$	*Haimon's Entrance*	0.56
$801\cdots805$	*Antigone's Entrance*	1.16
$817\cdots822$	*Chorus to Antigone*	0.57
$834\cdots838$	*Chorus to Antigone*	0.03
$929\cdots943$	*Chorus, Kreon and Antigone*	0.27
$1257\cdots1260$	*Chorus before Kreon's Kommos*	0.00
$1347\cdots1353$	*Final anapaests of the Chorus*	0.36
	mean $\omega = 0.40$, sdev $= 0.35$	

Table 4: Anapaests.

the same for anapaests. Lastly, Table 5 gives selections of dialogue (which is in iambic trimeters), divided between medium-to-long speeches and stichomythia.

As can be seen from the data, lyrics have the highest degree of non-projectivity, followed by speeches, then anapaests, and then stichomythia. To try and understand why this is the case, it will be useful to discuss Greek hyperbaton in my general terms.

Whereas in prose, hyperbaton corresponds to *strong focus*, which "does not merely fill a gap in the addressee's knowledge but additionally evokes and excludes alternatives" (Devine & Stephens 303), hyperbaton in verse only entails weak focus, which emphasizes but does not exclude (ibid. 107).

As a result, hyperbaton in verse may be used to evoke a kind of elevated style without incidentally entailing more emphasis and other pragmatic effects than intended. And so it should not be surprising that lyric passages, which reside in the most poetic and elevated register present in tragic diction, should have proved in *Antigone* to have the highest proportion of projectivity violations.

Within the lyric passages, the laments appear to have consistently higher ωs than the choral odes, which may stem from their being much more emotive and personal in nature. It should be noted that, whilst the odes are very tightly centered around the mean, there is a fair degree of variation in the ω for the laments.

Likewise, the anapaests vary so wildly in their ωs that it may be difficult to say very much about them that is relevant to the questions we are considering.

As for dialog, longer-form speeches are tightly wrapped around their mean, with

(a) Speeches and Dialogue

Lines		ω
162 · · · 210	Kreon: *ἄνδρες, τὰ μὲν δὴ...*	0.31
249 · · · 277	Guard: *οὐκ οἶδ'· ἐκεῖ γὰρ οὔτε...*	0.42
280 · · · 314	Kreon: *παῦσαι, πρὶν ὀργῆς...*	0.45
407 · · · 440	Guard: *τοιοῦτον ἦν τὸ πρᾶγμ'...*	0.50
450 · · · 470	Antigone: *οὐ γάρ τί μοι Ζεὺς...*	0.43
473 · · · 495	Kreon: *ἀλλ' ἴσθι τοι...*	0.56
639 · · · 680	Kreon: *οὕτω γὰρ, ὦ παῖ...*	0.50
683 · · · 723	Haimon: *πάτερ, θεοὶ φύουσιν...*	0.43
891 · · · 928	Antigone: *ὦ τύμβος, ὦ νυμφεῖον...*	0.37
998 · · · 1032	Teiresias: *γνώσῃ, τέχνης σημεῖα...*	0.42
1033 · · · 1047	Kreon: *ὦ πρέσβυ, πάντες...*	0.22
1064 · · · 1090	Teiresias: *ἀλλ' εὖ γέ τοι...*	0.77
1155 · · · 1172	Messenger: *Κάδμου πάροικοι καὶ...*	0.40
1192 · · · 1243	Messenger: *ἐγώ, φίλη δέσποινα...*	0.34
	mean ω = 0.44, sdev = 0.13	

(b) Stichomythia

Lines		ω
536 · · · 576	*Ismene, Antigone and Kreon*	0.26
728 · · · 757	*Haimon and Kreon*	0.33
991 · · · 997	*Kreon and Teiresias*	0.63
1047 · · · 1063	*Kreon and Teiresias*	0.10
1172 · · · 1179	*Chorus and Messenger*	0.21
	mean ω = 0.31, sdev = 0.20	

Table 5: Dialogue (Trimeters)

stichomythias varying a bit more. Speeches are a somewhat less projective than the stichmythias, being typically more eloquent and long-winded than their argumentative, choppy counterparts.

So far, the most surprising thing about the data is the degree to which certain passages vary in ω (or, if you like, the degree to which some passages *don't* vary in ω). The data draw us, then, to the following conclusions:

1. Non-projectivity varies within a single metrical type: that is, though lyric passages are in general less projective than anything else, some laments reach a degree of non-projectivity that exceeds the most elliptical odes in *Antigone*. Further, within the iambic trimeters, speeches are less projective than stichomythias.

2. Certain registers seem to be more conventionalized with respect to frequency of hyperbaton than others; that is, choral odes and speeches do not vary greatly amongst themselves, but laments and anapaests do.

It would then seem that meter itself is not a primary factor for predicting incidence of hyperbaton, but rather a secondary one only. That is to say, we know for a fact that passages in lyric meters have greater ω than passages in other meters. Yet, the variation of ω within that very meter indicates that there is some other factor involved, which very likely has to do with register along two different dimensions, which is to say, relative dignity and emotive force.

Appendices

A Algorithm & Data Representation

Dependency trees are a recursive data structure with a head node, which may have any number of arcs drawn to further trees (this is called a *rose tree*). We represent them as a Haskell data-type as follows:

$$\textbf{data}\ \mathsf{Tree}\ \alpha = \alpha \curvearrowright [\mathsf{Tree}\ \alpha]$$

This can be read as "For all types α, a Tree of α is constructed from a *label* of type α and a *subforest* of Trees of α," where brackets are a notation for lists.

Given a tree, we can extract its root label or its subforest by pattern matching on its structure as follows:

$$\begin{aligned}
&\mathsf{getLabel} :: \mathsf{Tree}\ \alpha \to \alpha \\
&\mathsf{getLabel}\ (l \curvearrowright _) = l \\
&\mathsf{getForest} :: \mathsf{Tree}\ \alpha \to [\mathsf{Tree}\ \alpha] \\
&\mathsf{getForest}\ (_ \curvearrowright ts) = ts
\end{aligned}$$

A.1 From Edges to Trees

We shall consider each word index to be a *vertex*, and each pair of vertices to be an Edge, which we shall write as follows:

$$\textbf{data}\ \mathsf{Edge}\ \alpha = \alpha \leftrightarrow \alpha\ \textbf{deriving}\ \mathsf{Eq}$$

An Edge α is given by two vertices of type α; the **deriving** Eq statement generates the code that is necessary to determine whether or not two Edges are equal using the $(\equiv)$ operator. In order to perform our analysis, we should wish to transform the raw list of edges into a tree structure. The basic procedure is as follows:

First, we try to find the root vertex of the tree. This will be a vertex that is given as the head of one of the words, but does not itself appear in the sentence:

$$\begin{aligned}
&\mathsf{rootVertex} :: \mathsf{Eq}\ \alpha \Rightarrow [\mathsf{Edge}\ \alpha] \to \mathsf{Maybe}\ \ \alpha \\
&\mathsf{rootVertex}\ es = \mathsf{find}\ (\notin \mathsf{deps})\ \mathsf{heads}\ \textbf{where} \\
&\quad \mathsf{heads} = [\![\ (\lambda(x \leftrightarrow y) \to x)\ es\ ]\!] \\
&\quad \mathsf{deps}\ \ = [\![\ (\lambda(x \leftrightarrow y) \to y)\ es\ ]\!]
\end{aligned}$$

If the data that we are working with are not well-formed, there is a chance that we will not find a root vertex; that is why the type is given as Maybe.

Then, given a root vertex, we look to find all the edges that it touches, and try to build the subtrees that are connected with those edges.

```
onEdge :: Eq α ⇒ α → Edge α → 𝔹
onEdge i (x ↔ y) = x ≡ i ∨ y ≡ i
oppositeVertex :: Eq α ⇒ α → Edge α → α
oppositeVertex i (x ↔ y)
  | x ≡ i       = y
  | otherwise = x
```

This is done recursively until the list of edges is exhausted and we have a complete tree structure:

```
treeFromEdges :: Ord α ⇒ [Edge α] → Maybe (Tree α)
treeFromEdges es = ⟦ (buildWithRoot es) (rootVertex es) ⟧ where
  buildWithRoot es root = root ↷ sortedChildren where
    roots           = ⟦ (oppositeVertex root) localVertices ⟧
    children        = ⟦ (buildWithRoot foreignVertices) roots ⟧
    localVertices   = filter (onEdge root) es
    foreignVertices = filter (¬ ∘ onEdge root) es
    sortedChildren  = sortBy (compare `on` getLabel) children
```

A.2 Counting Violations: Computing ω

Violations are given as an integer tally:

```
type Violations = ℤ
```

The basic procedure for counting projectivity violations is as follows: flatten down the tree into a list of edges cross-referenced by their vertical position in the tree; then traverse the list and see how many times these edges intersect each other.

```
type Level = ℤ
```

The vertical position of a node in a tree is represented as its Level, counting backwards from the total depth of the tree. That is, the deepest node in the tree is at level 0, and the highest node in the tree is at level n, where n is the tree's depth.

```
levels :: Tree α → [[α]]
levels t = fmap (fmap getLabel) $
  takeWhile (¬ ∘ null) $
  iterate (>>=getForest) [t]

depth :: Tree α → ℤ
depth = length ∘ levels
```

We can now annotate each node in a tree with what level it is at:

```
annotateLevels :: Tree α → Tree (Level, α)
annotateLevels tree = aux (depth tree) tree where
  aux l (x ↷ ts) = (l, x) ↷ ⟦ (aux (l − 1)) ts ⟧
```

Then, we fold up the tree into a list of edges and levels:

```
allEdges :: Ord α ⇒ Tree α → [(Level, Edge α)]
allEdges tree = aux (annotateLevels tree) where
  aux ((_, x) ↷ ts) = ts >>= go where
    go t@((l, y) ↷ _) = (l, edgeWithRange [x, y]) : aux t

edgeWithRange :: Ord α ⇒ [α] → Edge α
edgeWithRange xs = minimum xs ↔ maximum xs
```

A handy way to think of edges annotated by levels is as a representation of the arc itself, where the vertices of the edge are the endpoints, and the level is the height of the arc. Now, we can count the violations that occur between two arcs.

```
checkEdges :: Ord α ⇒ (Level, Edge α) → (Level, Edge α) → Violations
checkEdges (l, xy@(x ↔ y)) (l', uv@(u ↔ v))
  | x ∈E uv ∧ ((y ⩾ v ∧ l > l') ∨ y > v)  = 1
  | y ∈E uv ∧ ((x ⩽ u ∧ l > l') ∨ u < u)  = 1
  | u ∈E xy ∧ ((v ⩾ y ∧ l < l') ∨ v > y)  = 1
  | v ∈E xy ∧ ((u ⩽ x ∧ l < l') ∨ u < x)  = 1
  | x ∈E uv ∧ y ∈E uv ∧ l ⩾ l'            = 2
  | u ∈E xy ∧ v ∈E xy ∧ l ⩽ l'            = 2
  | otherwise                             = 0
```

We determine whether a vertex is in the bounds of an edge using $\cdot \in_E \cdot$.

$$\cdot \in_E \cdot :: \mathsf{Ord}\ \alpha \Rightarrow \alpha \to \mathsf{Edge}\ \alpha \to \mathbb{B}$$
$$z \in_E x \leftrightarrow y = z > \mathsf{minimum}\ [x, y]$$
$$\wedge\ z < \mathsf{maximum}\ [x, y]$$

We can now use what we've built to count the intersections that occur in a collection of edges. This is done by adding up the result of checkEdges of the combination of each edge with the subset of edges which are at or below its level:

```
edgeViolations :: Ord α ⇒ [(Level, Edge α)] → Violations
edgeViolations xs = sum ⟦ violationsWith xs ⟧ where
  rangesBelow (l, _) = filter (λ(l', _) → l' ⩽ l) xs
  violationsWith x   = sum ⟦ (checkEdges x) (rangesBelow x) ⟧
```

Finally, ω is computed for a tree as follows:

$$\omega :: \mathsf{Ord}\ \alpha \Rightarrow \mathsf{Tree}\ \alpha \to \mathbb{Q}$$
$$\omega\ tree = \frac{\mathsf{edgeViolations\ edges}}{\mathsf{length\ edges}}\ \textbf{where}$$
$$\mathsf{edges} = \mathsf{allEdges}\ tree$$

B Parsing the Perseus Treebank

We can express the general shape of a treebank document as follows:

```
type Document = [Sentence]
data Sentence = Sentence {sentenceId :: ℤ, sentenceEdges :: [Edge ℤ]}
```

To construct a Document from the contents of an XML file, it suffices to find all of the sentences.

```
documentFromXML :: [Content] → Document
documentFromXML xml = catMaybes ⟦ sentenceFromXML elems ⟧ where
  elems = onlyElems xml ⩥ findElements (simpleName "sentence")
```

Sentences are got by taking the contents of their `id` attribute, and extracting edges from their children.

```
sentenceFromXML :: Element → Maybe Sentence
sentenceFromXML e = ⟦ Sentence (readAttr "id" e) (pure edges) ⟧ where
```

```
  edges    = catMaybes ⟦ edgeFromXML children ⟧
  children = findChildren (simpleName "word") e
```

An edge is got from an element by taking the contents of its `id` attribute with the contents of its `head` attribute.

```
edgeFromXML :: Element → Maybe (Edge ℤ)
edgeFromXML e =
  case findAttr (simpleName "form") e of
    Just x | x ∈ [".", ",", ";", ":"] → Nothing
    otherwise → ⟦ (readAttr "head" e) ↔ (readAttr "id" e) ⟧
```

Thence, turn a sentence into a tree by its edges using the machinery from Section A.1.

```
treeFromSentence :: Sentence → Maybe (Tree ℤ)
treeFromSentence (Sentence _ ws) = treeFromEdges ws
```

By applying treeFromSentence to every sentence within a document, we can generate all the trees in a document.

```
treesFromDocument :: Document → [Tree ℤ]
treesFromDocument ss = catMaybes ⟦ treeFromSentence ss ⟧
```

By combining the above, we also may derive a document structure from a file on disk.

```
documentFromFile :: FilePath → IO Document
documentFromFile path = ⟦ (documentFromXML ∘ parseXML) (readFile path) ⟧
```

C Analysis of Data

We compute the mean ω of the trees contained in a document as follows:

```
analyzeDocument :: Document → ℚ
analyzeDocument doc = mean ⟦ ω (treesFromDocument doc) ⟧
```

We will wish to compare the ω for parts of *Antigone*. A section is given by a two sentence indices (a beginning and an end):

```
data Section = ℤ ··· ℤ
```

Then, the entire document can be cut down into smaller documents by section:

```
restrictDocument :: Section → Document → Document
restrictDocument (start ··· finish) = filter withinSection where
  withinSection (Sentence i _) = i ⩾ start ∧ i ⩽ finish
```

Auxiliary Functions

```
simpleName :: String → QName
simpleName s = QName s Nothing Nothing

readAttr :: Read α ⇒ String → Element → Maybe α
readAttr n = fmap read ∘ findAttr (simpleName n)

mean :: Fractional n ⇒ [n] → n
mean = ⟦ sum / length ⟧

sdev :: Floating n ⇒ [n] → n
sdev xs = √( sum ⟦ (λx → x²) ⟦ (−(mean xs)+) xs ⟧ ⟧ / (length xs − 1) )
```